
PROBLEM: 
A deformable C-shaped structure under gravity load undergoes an internal explosion. 
MESH: 
The fluid is meshed by Fl24 elements, the deformable structure by ED01 elements. 
The calculation is plane strain (DPLA) and is declared of the ALE type.  
MATERIALS: 
The air is modelled by the FLUT material (perfect gas), the explosive is solid TNT 
and the structure uses a linear elastic material (VM23). 
BOUNDARY CONDITIONS: 
The structure is clamped at the base. FSA fluid-structure interaction is used along the 
whole fluid-structure interface. 
LOADING: 
Gravity loading in the vertical direction is imposed on the whole model. 
OPTIONS: 
The QUAS STAT option is used to introduce an initial dynamic relaxation phase 
which allows to take into account approximately the static stresses generated in the 
structure by the gravity. This phase goes from t = -500 ms to t = 0. Then follows the 
explosion and the usual transient calculation. From t = 500 ms to the final time t = 1 s 
a second dynamic relaxation phase is introduced, which damps out the elastic 
oscillations of the system due to the explosion, and allows to compute the final, quasi-
static configuration of the system. 
CALCULATIONS: 
Preliminary calculations involving only the structure are used to estimate the system’s 
frequency under static loading (gravity). Then follows the real calculation as 
described in the previous point. 
RESULTS: 
The initial and final oscillations are effectively damped out by the quasi-static option. 

POST-TREATMENT 
Animations are produced. 
Numerical Solutions 
QUAS01 
Preliminary calculation modelling only the structure under gravity load and without 
dynamic relaxation, used to estimate the system’s oscillation frequency. The Cast3m 
mesh generation file reads: 
 
*%siz 100 
opti echo 1; 
* 
opti titr 'QUAS - 01'; 
opti dime 2 elem qua4; 
* 
p1p=0 0; 
p2p=10 0; 
p3p=10 5; 
p4p=0 5; 
pc = 5 5; 
tol=0.001; 
s1 = p2p d 5 p3p; 
s2 = p3p d 10 p4p; 
s3 = p4p d 5 p1p; 
stru = s1 et s2 et s3; 
elim tol (stru et pc); 
bloqall = stru poin droi p1p p2p tol; 
* 
mesh=stru et bloqall; 
* 
tass mesh; 
* 
opti sauv form 'quas01.msh'; 
sauv form mesh; 
opti trac psc ftra 'quas01_mesh.ps'; 
trac qual mesh; 
fin; 
 
The EUROPLEXUS input file reads: 
 
QUAS - 01 
$ 
 ECHO 
$VERI 
!CONV win 
CAST MESH 
DPLA NONL LAGR 
$ 
DIME 
 PT3L 21 ED01 20 ZONE 1 
TERM 
$ 
GEOM 
  ED01 stru 

TERM 
$ 
COMP EPAI 0.1 LECT stru TERM 
$ 
MATE 
$ structure 
     VM23 RO 2000 YOUN 3.E10 NU 0.2 ELAS 3.E10 
          TRAC 1 3.E10 1.0 
          LECT stru TERM 
$ 
LINK COUP 
     BLOQ 123 LECT bloqall TERM 
$ 
CHAR CONS GRAV 0 -9.80665D0 LECT tous TERM 
$ 
ECRI VITE ECRO TFRE 10.E-3 
     FICH K200 TFRE 2.0E-3 POIN TOUS 
               VARI VITE ECRO ECRC LECT 1 2 TERM 
     fich      alic temp tfre 1.e-3 
               poin lect pc term 
     fich      alic      tfre 1.e-3 
$ 
OPTI NOTE 
!QUAS STAT 100. 1. UPTO 0. 
 csta 0.8e0 
 log 1 
CALCUL TINI 0. TEND 500.E-3 pas1 1.e-5 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR  1 'dy_pc'   DEPL COMP 2 NOEU LECT pc TERM 
* 
trac 1 axes 1.0 'DISPL. [M]' 
LIST 1 axes 1.0 'DISPL. [M]' 
*================================================================= 
FIN 

 
By plotting the vertical displacement of the central point of the structure one estimates 
an oscillation period under gravity of about 0.33 s (i.e. a frequency of 3 Hz): 

 
QUAS02 
Further preliminary calculation modelling only the structure under gravity load with 
dynamic relaxation, to verify that the computed system frequency is correct: 
The EUROPLEXUS input file reads: 
 
QUAS - 02 
$ 
 ECHO 
$VERI 
!CONV win 
CAST 'quas01.msh' MESH 
DPLA NONL LAGR 
$ 
DIME 
 PT3L 21 ED01 20 ZONE 1 
TERM 
$ 
GEOM 
  ED01 stru 
TERM 
$ 
COMP EPAI 0.1 LECT stru TERM 
$ 
MATE 
$ structure 

     VM23 RO 2000 YOUN 3.E10 NU 0.2 ELAS 3.E10 
          TRAC 1 3.E10 1.0 
          LECT stru TERM 
$ 
LINK COUP 
     BLOQ 123 LECT bloqall TERM 
$ 
CHAR CONS GRAV 0 -9.80665D0 LECT tous TERM 
$ 
ECRI VITE ECRO TFRE 10.E-3 
     FICH K200 TFRE 2.0E-3 POIN TOUS 
               VARI VITE ECRO ECRC LECT 1 2 TERM 
     fich      alic temp tfre 1.e-3 
               poin lect pc term 
     fich      alic      tfre 1.e-3 
$ 
OPTI NOTE 
 QUAS STAT 3. 1. !UPTO 0. 
 csta 0.8e0 
 log 1 

CALCUL TINI 0. TEND 500.E-3 pas1 1.e-5 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 

AXTE 1.0 'Time [s]'  
* 
COUR  1 'dy_pc'   DEPL COMP 2 NOEU LECT pc TERM 
RCOU  2 'dy_pc'   FICH 'quas01.pun' RENA 'dy_pc_no_damp' 
* 
trac 2 1 axes 1.0 'DISPL. [M]' yzer 
     COLO noir roug 
*================================================================= 
FIN 

By comparing the results without and with initial dynamic relaxation one has: 
 

 
The damping is indeed critical. In agreement with the linear theory, the “static” 
displacement is one half of the dynamic displacement. 
 
QUAS10 
Final calculation modelling the complete system (fluid and structure) and containing 
an initial and a final dynamic relaxation phases, as explained above. 
The Cast3m mesh generation file reads: 
 
*%siz 100 
opti echo 1; 
* 
opti titr 'QUAS - 10'; 
opti dime 2 elem qua4; 
* 
p1=0 0; 
p2=10 0; 
p3=10 5; 
p4=0 5; 
pb= 5 2; 
pc= 5 5; 
tol=0.001; 
c1 = p1 d 10 p2; 
c2 = p2 d 5 p3; 
c3 = p3 d 10 p4; 
c4 = p4 d 5 p1; 
flui = dall c1 c2 c3 c4 plan; 
elim tol (flui et pb et pc); 
bull = flui elem appu larg pb; 
air = diff flui bull; 
p1p = p1 plus p1; 
p2p = p2 plus p1; 
p3p = p3 plus p1; 
p4p = p4 plus p1; 

s1 = p2p d 5 p3p; 
s2 = p3p d 10 p4p; 
s3 = p4p d 5 p1p; 
stru = s1 et s2 et s3; 
bloqall = stru poin droi p1 p2 tol; 
fsan = cont flui; 
fsan = chan poi1 fsan; 
list (nbno fsan); 
fsrn = flui poin droi p1 p2 tol; 
list (nbno fsrn); 
fsan = diff fsan fsrn; 
list (nbno fsan); 
* 
mesh=flui et stru et bloqall et fsan et fsrn; 
* 
tass mesh; 
* 
opti sauv form 'quas10.msh'; 
sauv form mesh; 
opti trac psc ftra 'quas10_mesh.ps'; 
*trac qual mesh; 
bull = bull coul roug; 
air = air coul turq; 
trac face (bull et air); 
fin; 

 

The EUROPLEXUS input file reads: 
 
QUAS - 10 
$ 
 ECHO 
$VERI 
!CONV win 
CAST MESH 
DPLA NONL ALE 
$ 
DIME 
 PT3L 21 PT2L 66 FL24 50 ED01 20 ZONE 2 
 NALE 87 NBLE 87 
TERM 
$ 
GEOM 
  FL24 flui ED01 stru 
TERM 
$ 
COMP EPAI 0.1 LECT stru TERM 
$ 
opti rezo mvre modu 
GRIL LAGR LECT stru TERM 
     EULE LECT fsan fsrn TERM 
     ALE  LECT flui TERM 
     AUTO autr 
$ 
MATE 
**   l'air :  on calcule eint pour avoir P=1 bar (P=omeg*ro*eint)  
    flut ro 1.3 eint 0.21978e6 gamm 1.35 PB 0 
         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
         CQ 2.56 PMIN 0 PREF 1.e5 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         ros 1630 
         LECT air TERM 
        !LECT air bull  TERM 
**  Le TNT :   on donne directement ro = ros 
*              avec ignition au point PDET a l'instant 0.0  
*              la vitesse de detonation est celle de Chapman-Jouguet 
    flut ro 1630 eint 3.68e6 gamm 1.35 PB 0 
         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
         CQ 2.56 PMIN 0 PREF 1.e5 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         d 6930 pini 1e5 
         TDET 0. 
         xdet 5.0 ydet 2.0 

         LECT bull TERM 
$ structure 
     VM23 RO 2000 YOUN 3.E10 NU 0.2 ELAS 3.E10 
          TRAC 1 3.E10 1.0 
          LECT stru TERM 
$ 
LINK COUP 
     BLOQ 123 LECT bloqall TERM 
     FSA      LECT fsan TERM 
     FSR      LECT fsrn TERM 
$ 
CHAR CONS GRAV 0 -9.80665D0 LECT tous TERM 
$ 
ECRI VITE ECRO TFRE 10.E-3 
     FICH K200 TFRE 2.0E-3 POIN TOUS 
               VARI VITE ECRO ECRC LECT 1 2 TERM 
     fich      alic temp tfre 1.e-3 
               poin lect pc term 
     fich      alic      tfre 1.e-3 
$ 
OPTI NOTE 
 QUAS STAT 3. 1. UPTO 0. FROM 500.E-3 
 csta 0.4e0 
 dtml 
 stel 
 log 1 
CALCUL TINI -500.E-3 TEND 1000.E-3 pas1 1.e-5 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR  1 'dy_pc'   DEPL COMP 2 NOEU LECT pc TERM 
* 
trac 1 axes 1.0 'DISPL. [M]' 
*================================================================= 
FIN 

The deflection of the central point is (note the initial and final relaxations): 

 
The velocities and pressures during the various phases are: 

 


